DOCUMENT RESUME

ED 081 206 EM 011 378

AUTHOR	Hafner, Everett
TITLE	Computers, Synthesizers, and the Physics of Music.
PUB DATE	Jun 73
NOTE	16p.; Paper presented at the Conference on Computers in the Undergraduate Curricula (Claremont, California, June 18-20, 1973)
EDRS PRICE	MF-$0.65 HC-$3.29
DESCRIPTORS	*Acoustics; Aesthetic Education; *Computer Assisted Instruction; Computers; Course Descriptions; Electronics; Experiments; Higher Education; Independent Study; *Music; Musical Composition; Music Appreciation; *Music Education; Music Theory; Perception; *Physics; Programing; Undergraduate Study
IDENTIFIERS	Computer Based Composition; Hampshire College Amherst Massachusetts; Lissajous Patterns; *Synthesizers

ABSTRACT

An experimental course for the study of acoustics is taught at Hampshire College for students interested in the interface of physics and music. This paper outlines the course, describes the basic techniques it employs, gives examples of successful exercises, and discusses some difficulties not yet overcome. Computers are heavily utilized, removing the need for mathematics prerequisites and clearing the way for the examination of relatively sophisticated problems. Synthesizers are employed to translate formulae into sounds and to create opportunities for broad experimentation in music theory, perception, and composition. Both techniques open up possibilities for independent study by students interested in programing, recording techniques, analysis of musical instruments, digital and analog electronics, computer-based composition, the aesthetics of Lissajous patterns, and other topics. (Author/PB)

Computers, Synthesizers, & the Physics of Music *

Everett Hafner
Hampshire College
Amherst
Mass.
01002

(413) 542-4779

Abstract

An experimental course has been developed for the study of acoustics at an elementary college level. It is most attractive to students whose interests include physics and music. The use of computers removes prerequisites in mathematics and clears the way for examination of relatively sophisticated problems. The use of synthesizers translates formulae into sounds and creates opportunities for broad experimentation in music theory, perception, and composition. Both techniques open up possibilities for independent study by students interested in programming, recording techniques, analysis of musical instruments, digital and analog electronics, computer-based composition, the aesthetics of Lissajous patterns, and other peripheral aspects of the course. This paper outlines the course, describes the basic techniques it employs, gives examples of successful exercises, and discusses some difficulties not yet overcome.

* contributed to the Fourth Conference on Computers in the Undergraduate Curriculum, Claremont, June 1973

"We also have Sound-houses, where we practise and demonstrate all Sounds and their Generation. We have Harmonies which you have not, of Quarter Sounds, and lesser Slides. . ."

Roger Bacon "The New Atlantis" 1624

The science of acoustics, both stimulated by and reflecting back upon the production of music, is so ancient and so honorable that we have tended, until very recently, to regard it as classical and closed. After astronomy and medicine, it is probably the oldest of all the sciences we know. Our understanding of musical instruments and scales is, in most essential ways, no deeper than the Greek philosophers', and it is now almost three and a half centuries since Galileo discovered the dependence of string frequencies on length, density and tension. After the work of Rayleigh (1877) very little was added to the subject until 1930, when physicists discovered the anomalous dispersion of sound in gases and sought its explanation in the quantum theory of specific heats.

Most of the current research in acoustics concentrates, as in the theory of phonons, on its important relation to solid state physics. Laboratory techniques, especially in cryogenics and ultrasonics, have grown rapidly to meet the needs of the new theories. But the classical body of knowledge has remained intact and almost untouched for a hundred years. Thus, during that period, development of conventional bowed and blown instruments has virtually ceased; our keyboards, and hence most of musical composition, have locked themselves into the twelve-tone tempered scale; and we still believe with Helmholtz that the nonlinearity of the ear produces perceptible sum and difference tones, despite the absence of data in support of the idea. Even the name of the science is badly used, as when musicians speak of "acoustic" instruments in distinguishing classical hand- or mouth-powered devices from their new electrically based companions.

Those of us who practice physics and music with equal dedication are glad to see a rebirth of activity in musical science. Several forces are strongly at work: among them are the traditional restlessness of artists in search of new ideas, a movement among physicists to return to tasks of broad interest, and the arrival of a vigorous young generation who see modern technology as more than a means of making money and waging war. But the largest force, perhaps, is the rapid progress of technology itself. Beginning with the invention of the vacuum tube, the history of electronics at every step is the record of new incentives for extending our understanding and control of musical sound.

In these circumstances it is not surprising that courses in the physics of music are growing up in schools and colleges everywhere. Since 1969, when it established one of the first collegiate electronic music studios, Hampshire College has been building ways of using synthesizers as laboratory tools for the study of physics. Concurrently, with the emergence of efficient time-shared computation and the growth of strong interest in APL as an effective pedagogical instrument, we have learned to exploit special computer techniques for handling problems that many students would otherwise have found difficult. Like all teachers of physics, we recognize the fact that students come to us with wide variety of training and experience in mathematics. Their approaches to the course can vary accordingly. Naive students can use the technique as a direct and simple tool for solving problems in physics; students with a working knowledge of mathematics can examine and edit our programs for their own purposes; advanced students can attack their tasks independently, and can assist us in the design of new programs.

Our course begins, as we believe all such studies must, with mathematical preliminaries. The following is a typical exercise in integration using Simpson's Rule:

> The APL function INT estimates the area under the graph of a function specified at N points. Use it to estimate the area under 1/X from X = 1 to N, for various N, with N points. Compare the results with log X.

```
    ∇ Z←INT;N;X;F;K
[1]   'X VALUES?'
[2]   N←⍴X←⎕
[3]   →(0=2|N)/AGAIN
[4]   'FUNCTION?'
[5]   F←⎕
[6]   →((⍴F)≠N)/WRONG
[7]   K←1+Z←0
[8]   GO:Z←Z+(÷6)×(X[K+2]-X[K])×F[K]+F[K+2]+4×F[K+1]
[9]   →(N>K←K+2)/GO
[10]  →0
[11]  AGAIN:'CHOOSE AN ODD NUMBER OF VALUES.'
[12]  →2
[13]  WRONG:'F MUST HAVE A VALUE FOR EACH VALUE OF X.'
[14]  →5
```

```
      INT
X VALUES?
⎕:
      ⍳11
FUNCTION?
⎕:
      ÷X
 2.411
      ⍟11
 2.398
```

```
      INT
X VALUES?
⎕:
      ⍳101
FUNCTION?
⎕:
      ÷X
 4.628
      ⍟101
 4.615
```

```
      INT
X VALUES?
⎕:
      ⍳1001
FUNCTION?
⎕:
      ÷X
 6.922
      ⍟1001
 6.909
```

An early exercise in oscillatory motion:

> Use the function OSC to study simple harmonic motion by setting PWR ← 1, choosing reasonable time intervals for integration. Then try other values of PWR , varying the amplitude as well.

```
    ∇ A OSC DT;X;V;T;R
[1]   X←A
[2]   T←0×V←0.5×DT×X*PWR
[3]   R←0,A,A
[4]   X←X+(V←V-DT×X*PWR)×DT
[5]   →((2×T)≠⌊2×T←T+DT)/4
[6]   R←R,T,X,A×2○T
[7]   →(T<10)/4
[8]   (((⍴R)÷3),3)⍴R
```

PWR←1

1 OSC .125

0	1	1
0.5	0.8774	0.8776
1	0.5398	0.5403
1.5	6.976E¯2	7.074E¯2
2	¯0.4173	¯0.4161
2.5	¯0.8021	¯0.8011
3	¯0.9903	¯0.99
3.5	¯0.9357	¯0.9365
4	¯0.6517	¯0.6536
4.5	¯0.2079	¯0.2108
5	0.2868	0.2837
5.5	0.7112	0.7087
6	0.9613	0.9602
6.5	0.9757	0.9766
7	0.7509	0.7539
7.5	0.342	0.3466
8	¯0.1507	¯0.1455
8.5	¯0.6064	¯0.602
9	¯0.9135	¯0.9111
9.5	¯0.9967	¯0.9972
10	¯0.8355	¯0.8391

PWR←÷3

1 OSC .125

0	1	1
0.5	0.8758	0.8776
1	0.5145	0.5403
1.5	¯4.114E¯2	7.074E¯2
2	¯0.5687	¯0.4161
2.5	¯0.8937	¯0.8011
3	¯0.9799	¯0.99
3.5	¯0.8195	¯0.9365
4	¯0.4273	¯0.6536
4.5	0.1428	¯0.2108
5	0.6503	0.2837
5.5	0.9449	0.7087
6	0.9959	0.9602
6.5	0.799	0.9766
7	0.3726	0.7539
7.5	¯0.213	0.3466
8	¯0.6953	¯0.1455
8.5	¯0.9593	¯0.602
9	¯0.9786	¯0.9111
9.5	¯0.7513	¯0.9972
10	¯0.2996	¯0.8391

In doing this, students must know enough of APL to understand the syntax of the function, the meaning of A and DT, and the way in which the program sets initial conditions. They must also know that the third column tabulates cos T and they can simply observe that the linear case matches that solution. The form of the function makes the transition to non-linear motion easy and transparent. With slight editing, the same program can yield the dependence of the period of a simple pendulum on its amplitude; by plotting results, a student can discover a formula for the second-order correction to the period for small amplitude.

We approach the study of waves on continuous systems by examining lattices of N identical springs joining N - 1 identical masses, with the ends of the system fixed. The APL function for this is:

```
    ∇ M LATTICE C;N;DT;T;V;Y;Z;A
[1]    ''
[2]    V←(1+N←C[1])⍴T←0
[3]    W←Y←0,1○(○M)×(⍳N)÷N
[4]    GO:Z←(1⌽Y)+(¯1⌽Y)-2×Y
[5]    Z[1]←Z[N+1]←0
[6]    Y←Y+DT×V←V+Z×(DT←C[3]÷M)×N×(N-1)
[7]    →(~∨/(T←T+DT)=A←(0.5÷M)×⍳C[2]÷0.5)/AHEAD
[8]    W←W,Y
[9]    →GO
[10]   AHEAD:→(T≤C[2]÷M)/GO
[11]   W←⍉(((⍴W)÷N+1),N+1)⍴W
[12]   W←0,A,,⌊0.5+100×W
[13]   ⍉((N+2),((⍴W)÷N+2))⍴W
[14]   ''
```

The program is arranged so that, for a system of fixed length, the tension and total mass are the same for all N. It sets up an initial sinusoidal displacement in the Mth mode and tabulates the motion of each mass thenceforth. From his graphs of the results, a student can estimate the period for each normal mode of a given system and compare it with theory. He observes the fact that, when N is finite, the spectrum of frequencies is not harmonic. But as N grows large the system tends toward a continuous string whose modes are well known. In our case the limiting spectrum is M/2 and this fact becomes apparent for the low modes when N exceeds 10 or so. Handled in this elementary way (the program "knows" no physics beyond Newton's Second Law and almost no mathematics at all), the problem can be studied without any of the usual difficulties.

Discussion of waves in two and three dimensions follows, with the usual treatment of reflection, dispersion, interference and diffraction. Sound waves are a natural resource for all such studies, and it is at this stage that we introduce students to the synthesizer. We work principally with the Synthi-AKS, a compact instrument (Fig. 1) that we find ideal for demonstration and student use. It contains three function generators, a white noise source, ring modulation, facilities for filtering and other forms of treatment, and a 256-event digital sequencer for programming





Fig. 1 The Synthi-AKS Synthesizer

musical patterns. Modules of the Synthi are conveniently interconnected by inserting pins in a 16 x 16 matrix. Our work requires, in addition to this instrument, a triggered oscilloscope with camera, a frequency counter, a standard laboratory function generator, and a two-channel recording and reproducing audio system of good quality.

Needless to say, the sudden opportunity for free access to this laboratory generates and reinforces many keen interests. Frequent distractions from the thread of the course are difficult to resist. There is nevertheless a sequence of systematic experiments and computations that a student is expected to complete. He begins by mastering the synthesizer, following classroom discussion of its principal components. The oscilloscope is used throughout for observation of waveforms and their modifications. Of the two examples shown here, the first is a triangular wave before and after filtering with emphasis on its seventh harmonic. The second is a display of the sum and the product of sine waves at 200 hz and 4400 hz; the latter treatment, provided by the ring modulator of







the Synthi, yields components at 4200 and 4600 hz. The student eventually sees and hears all of the characteristic "electronic" sounds. He may also, of course, observe waveforms produced by classical instruments; the traces above were taken from the open 83.5-hz string of a guitar at 0.2 and 0.5 sec after pluck. (The larger and earlier trace is almost pure third harmonic.)

The synthesizer was designed for composers; yet it lends itself, often fortuitously, to many experiments in physics. For example its internal speakers are about one wavelength apart at 1000 hz, making the observation of interference patters easy when the two channels are driven by a single tone. Similarly, we can measure the speed of sound by several methods, the limits of human perception of pitch differences (using the keyboard, whose intervals can be made arbitrarily small), the phenomena of frequency modulation, and the resonant frequencies of a violin (in a feedback loop consisting of violin, contact microphone, variable bandpass filter, and speaker). We now suspect, in fact, that the list of useful things to do is endless.

We devote a large amount of effort, naturally, to the study of Fourier analysis. It is wise for most students to begin by observing convergence of simpler sequences. Here is an APL function for looking quickly at Newton's iteration for square roots:

```
      ∇ Y SQR X                                                     29*.5
[1]     →((|(!⎕←Y←0.5×Y+X÷Y)-X*0.5)>10*¯8)/1                 5.385164807

                                   ¯5 SQR 29                       5 SQR 29
       1.4 SQR 2               ¯5.4                          5.4
     14285714                  ¯5.385185185                  5.385185185
     14213564                  ¯5.385164807                  5.385164807
```

After discussion of series and convergence, and after introducing the Fourier theorem qualitatively, we observe harmonic spectra using the sources and filter of the Synthi. Students rapidly see such things as the absence of even harmonics in triangular and square waves, the precise integer ratios of harmonic frequencies (using the frequency counter), and the dependence of richness on types of discontinuity (the sine wave is the limiting case of highest continuity and lowest richness). Then we assign the following:

> Use the function FOURIER to investigate the spectra of N-point samples of sine, triangle, square, sawtooth and pulse waveforms. Develop a rule for the optimum number of harmonics for a given N. Analyze other waves of your choice.

The function is:

```
    ∇ Z←FOURIER M;F;X;N;K;D
[1]    →(0=2|N←ρF←⎕)/1
[2]    X←(÷¯1+N)×(○2)×⍳N
[3]    A←((M+1),3)ρK←0
[4]    A[;0]←⍳1+M
[5]    A[0;1]←(÷○2)×(D←0,(○2),N) INT F
[6]   RUN:→(M<K←K+1)/PRINT
[7]    A[K;1]←(÷○1)×D INT F×2○K×X
[8]    A[K;2]←(÷○1)×D INT F×1○K×X
[9]    →RUN
[10]  PRINT:COEF←0.0001×⌊0.5+10000×A
[11]   Z←(N,3)ρK←0
[12]  OUT:Z[K;1]←(+/A[;1]×2○(⍳1+M)×X[K])++/A[;2]×1○(⍳1+M)×X[K]
[13]   Z[K;2]←Z[K;1]-Z[K;0]←F[K]
[14]   →(N>K←K+1)/OUT
[15]   FVAL←Z←0.0001×⌊0.5+10000×Z
[16]   Z←(+/Z[;2]*2)÷N
```

```
    ∇ Z←X INT F;K
[1]    K←Z←0
[2]    Z←Z+((X[1]-X[0])÷3×¯1+X[2])×F[K+2]+F[K]+4×F[K+1]
[3]    →(X[2]>2+K←K+2)/2
```

The argument M of FOURIER is the highest harmonic to which the analysis will be carried. The function calls for an input sample and yields the mean square deviation of the fitted curve from the input. It stores tables of coefficients and functional values. Here is the fit to 21 samples of a rising sawtooth:

```
      FOURIER 5
⎕:
      SAW 21
0.1006
      COEF
    0          0          0
    1          0          ¯0.6367
    2          0          ¯0.3186
    3          0          ¯0.2132
    4          0          ¯0.1619
    5          0          ¯0.1333
```

```
      FVAL[⍳11;]
¯1          0          1
¯0.9        ¯0.8438    0.0562
¯0.8        ¯0.9752    ¯0.1752
¯0.7        ¯0.6555    0.0445
¯0.6        ¯0.5135    0.0865
¯0.5        ¯0.5567    ¯0.0567
¯0.4        ¯0.4468    ¯0.0468
¯0.3        ¯0.2398    0.0602
¯0.2        ¯0.1789    0.0211
¯0.1        ¯0.1614    ¯0.0614
0           0          0
```

It is left for the student to compare the sine coefficients to the predictions of a conventional mathematical treatment, which this method avoids. He can use the technique on many important problems. For example, it is an interesting fact that a parabolic arc is a good fit to half of a sine wave; two such arcs join to fit the whole wave. How good is the fit in the Fourier sense?

```
    ∇ Z←PAR N;X
[1]   Z←X,(-X←(4×X÷○1)×1-(X←((○2)÷N-1)×⍳(N-1)÷2)÷○1),0
    ∇

      FOURIER 5
⎕:
      PAR 21
7.037E¯6
      COEF
   0          0          0
   1          0          1.032
   2          0          0
   3          0          0.0389
   4          0          0
   5          0          0.0107
```

indicates that the fit is excellent. Since the function generators of the Synthi produce their sine waves by integrating triangles, the lesson is appropriate.

We devote the rest of our time to the theory of music and its composition. Having listened to exact and tempered intervals on the Synthi, we carry out the following:

> Compute the diatonic scale based on middle C (261.63 hz) and compare it with appropriate steps in tempered scales of 12, 19 and 31.

```
      DI←1,(9÷8),(10÷9),(16÷15),(9÷8),(10÷9),(9÷8),16÷15

      ×/DI
2
      DI
1  1.125  1.111  1.067  1.125  1.111  1.125  1.067

      DS←1,DI[2],(×/DI[⍳3]),(×/DI[⍳4]),(×/DI[⍳5]),×/DI[⍳6]
      DS←DS,(×/DI[⍳7]),×/DI

      DS
1  1.125  1.25  1.333  1.5  1.667  1.875  2

      CA←440
      CC←.5×CA×T12[4]

 T12←1,2*(⍳12)÷12
 T19←1,2*(⍳19)÷19
 T31←1,2*(⍳31)÷31

 X←DS,T12[1 3 5 6 8 10 12 13],T19[1 4 7 9 12 15 18 20]
 X←X,T31[1 6 11 14 19 24 29 32]

 ⍉4 8⍴CC×X

261.63        261.63        261.63        261.63
294.33        293.66        291.88        292.57
327.03        329.63        325.64        327.18
348.83        349.23        350.29        349.88
392.44        392           390.81        391.27
436.04        440           436.01        437.55
490.55        493.88        486.43        489.3
523.25        523.25        523.25        523.25
```

The exercise is usefully extensible in several directions: to the theory of piano tuning, the design of frequency dividers for electronic organs, problems of just intonation (or "musimatics," as A.L.L. Silver has called its theory), and the acoustic significance of prime number scales such as T19, T31 and T53.

Passing further into the mathematics of music:

> Construct a function that will write a full peal of Plain Bob on N bells. How does the length of the peal depend on N?

```
    ∇ Z←PEAL ROUND
[1]   Z←ROUND
[2]   Z←Z,P1(-ρROUND)↑Z
[3]   Z←Z,P2(-ρROUND)↑Z
[4]   →(0=∧/ROUND=(-ρROUND)↑Z)/2
    ∇
      ∇P1[⎕]∇
    ∇ Z←P1 N
[1]   Z←,⌽(((ρN)÷2),2)ρN
    ∇
      ∇P2[⎕]∇
    ∇ Z←P2 N
[1]   Z←N[1],(P1 1↓¯1↓N),N[ρN]
    ∇

      RING←PEAL 15 14 12 10 8 7 5 3
      ρRING
136
      9 16ρRING
 15 14 12 10  8  7  5  3 14 15 10 12  7  8  3  5
 14 10 15  7 12  3  8  5 10 14  7 15  3 12  5  8
 10  7 14  3 15  5 12  8  7 10  3 14  5 15  8 12
  7  3 10  5 14  8 15 12  3  7  5 10  8 14 12 15
  3  5  7  8 10 12 14 15  5  3  8  7 12 10 15 14
  5  8  3 12  7 15 10 14  8  5 12  3 15  7 14 10
  8 12  5 15  3 14  7 10 12  8 15  5 14  3 10  7
 12 15  8 14  5 10  3  7 15 12 14  8 10  5  7  3
 15 14 12 10  8  7  5  3 15 14 12 10  8  7  5  3

      PRINT 2 24ρ(⍳24),RING[⍳24]

F  | +  +  +  + | +  +  +  + | +  +  +  + | +  +  +  + | +  +  +  + | +  +  +  + |
   | ⎕          |            |    ⎕       |            |       ⎕    |            |
D  | + ↑⎕ +  +  | +  +  +  + |↑⎕  +  +  + | +  +  +  + |↑⎕  +  +  + | +  +  +  + |
   |      ↑⎕    |            |         ↑⎕ |            |            |↑⎕          |
B  | +  +  +  ⎕ | +  +  +  + | +  +  ⎕  + | +  +  +  + | +  ⎕  +  + | +  +  +  + |
   |            | ⎕          |            |    ⎕       |            |       ⎕    |
G  | +  +  +  + | + ↑⎕ +  +  | +  +  +  + |↑⎕  +  +  + | +  +  + ↑⎕ | +  +  +  + |
   |            |      ↑⎕    |            |         ↑⎕ |            |         ↑⎕ |
   | +  +  +  + | +  +  +  ⎕ | +  +  +  + | +  +  ⎕  + | +  +  +  + | +  ⎕  +  + |
```

It is not well known that certain mathematical problems in change ringing, falling naturally into group theory, are unsolved. For example there are 5040 changes possible on seven bells. Can they all be reached in a peal of Stedman Triples with Plains and Bobs alone? Nobody knows. Similar questions might arise in a mathematical theory of tonerows, if it were to develop beyond its present primitive stage. In our course we propose four operators on tonerows, in extension of the familiar forms of retrograding and mirroring:

```
    ∇ Z←NEWROW
[1]   ROW←Z← 2 12 ρ(Z[⍋Z←12?24]),LIM ¯3+(12?12)+12×¯1+?12ρ2
    ∇ Z←REP X
[1]   VROW←Z← 2 12 ρX[1;],LIM 20-X[2;]
    ∇ Z←RET X
[1]   VROW←Z← 2 12 ρ(Z[⍋Z←25-X[1;]]),⌽X[2;]
    ∇ Z←MRP X
[1]   VROW←Z← 2 12 ρX[1;],LIM(2×X[2;1])-X[2;]
    ∇ Z←MRT X
[1]   VROW←Z← 2 12 ρ(Z[⍋Z←⌊0.5+24|¯0.5+(2×X[1;1])-X[1;]]),X[2;]
    ∇ Z←LIM X
[1]   X←X+(12×X<1)-12×X>19
[2]   →((1>⌊/X)∨19<⌈/X)/1
[3]   Z←X
    ∇
      NEWROW
  2   4   5   9  11  12  13  18  19  20  23  24
  9  10  14  12  18   8  17  19  11   1  15   4
      REP ROW
  2   4   5   9  11  12  13  18  19  20  23  24
 11  10   6   8   2  12   3   1   9  19   5  16
      RET ROW
  1   2   5   6   7  12  13  14  16  20  21  23
  4  15   1  11  19  17   8  18  12  14  10   9
      MRP ROW
  2   4   5   9  11  12  13  18  19  20  23  24
  9   8   4   6  12  10   1  11   7  17   3  14
      MRT ROW
  2   4   5   8   9  10  15  16  17  19  23  24
  9  10  14  12  18   8  17  19  11   1  15   4
```

In each case, the first line denotes times (12 beats at random out of 24) and the second denotes the 12 pitches of the row. The obvious question concerns the possibility of forming a group with such a set of operators as elements.

We end our course with discussion of the role of computers in composition and synthesis. Since the Synthi-AKS contains a sequencer that is, in effect, a small computer memory, students can begin to study the long set of technical and aesthetic problems that stretch before us as computers enter the production of music in fundamentally new ways. The sequencer lends itself nicely to storage of a peal of bells, and it is interesting to wonder whether this ancient science, too, will gain new life as technology continues to unfold.

The principal difficulty in the course so far has been the consuming task of assembling adequate equipment, a full library of programs, and good ways of introducing both. We have found the Synthi admirable in accomplishing much of the first part of this task, and APL ideal in the second. An equally large difficulty has to do with establishing healthy attitudes among students who, left to themselves, tend to play endlessly and aimlessly with the admittedly fascinating resources in their hands. Our main job, after all, is to introduce them to physics, both as a pure science and as an implement for progress in music. However adept he may be at operating synthesizers, a student fails to meet our objectives if he leaves us still ignorant of the physical laws at hand.

We are, of course, happy to correspond with teachers who wish to have further details about this work.